# $Id$
# Maintainer: AndyRTR <andyrtr@archlinux.org>

pkgname=hyphen-de
pkgver=20060120
pkgrel=4
pkgdesc="German hyphenation rules"
arch=(any)
url="http://www.openoffice.org/lingucomponent/"
license=('LGPL')
optdepends=('hyphen: offers hyphenation library functions')
source=(#http://ftp.services.openoffice.org/pub/OpenOffice.org/contrib/dictionaries/hyph_de_DE.zip
        #http://ftp.services.openoffice.org/pub/OpenOffice.org/contrib/dictionaries/hyph_de_CH.zip
        # mirror from ApacheOpenOffice - see: http://www.openoffice.org/distribution/mirrors/master.html
        # http://sunsite.informatik.rwth-aachen.de/ftp/pub/mirror/OpenOffice/contrib/dictionaries/hyph_de_DE.zip
        # http://sunsite.informatik.rwth-aachen.de/ftp/pub/mirror/OpenOffice/contrib/dictionaries/hyph_de_CH.zip
        http://www.mirrorservice.org/sites/download.openoffice.org/contrib/dictionaries/hyph_de_DE.zip
        http://www.mirrorservice.org/sites/download.openoffice.org/contrib/dictionaries/hyph_de_CH.zip)
md5sums=('20e5b706ee3201aa2741dabca824cd8b'
         '7b8567cc0ee1623dded40f2cba366e91')

build() {
  /bin/true
}

package() {
  cd "$srcdir"
  install -dm755 ${pkgdir}/usr/share/hyphen
  cp -p hyph_de_??.* $pkgdir/usr/share/hyphen

  pushd $pkgdir/usr/share/hyphen/
  de_DE_aliases="de_AT de_BE de_LU"
  for lang in $de_DE_aliases; do
        ln -s hyph_de_DE.dic hyph_$lang.dic
  done
  de_CH_aliases="de_LI"
  for lang in $de_CH_aliases; do
        ln -s hyph_de_CH.dic hyph_$lang.dic
  done

  popd

  # the symlinks
  install -dm755 ${pkgdir}/usr/share/myspell/dicts
  pushd $pkgdir/usr/share/myspell/dicts
    for file in $pkgdir/usr/share/hyphen/*; do
      ln -sv /usr/share/hyphen/$(basename $file) .
    done
  popd
  
  # docs
  install -dm755 ${pkgdir}/usr/share/doc/$pkgname
  cp -p README_hyph_de_DE.txt $pkgdir/usr/share/doc/$pkgname
  cp -p README_hyph_de_CH.txt $pkgdir/usr/share/doc/$pkgname
}
